
Mplus syntax file to accompany multilevel structural equation model (MSEM) from: 
 
Preacher, K. J., Dunkley, D. M., & Zuroff, D. C. (January, 2010). Demonstrating multilevel 
structural equation modeling for testing mediation: Effects of self-critical perfectionism on daily 
affect. Talk given at the annual conference of the Society for Personality and Social Psychology, Las 
Vegas, NV. 
 
TITLE: Various MSEM models for Dunkley perfectionism data 
DATA: FILE IS dunkley_011110.dat; ! text file containing raw data in long format 
VARIABLE: NAMES ARE 
DAY ID AGE SEX LANGUAGE FACULTY YEAR LIVING EVENT1 EVENT2 EVENT3 
EVENT4 EVENT5 EVENT6 EVACAD EVSOC EVGEN MPSSLF1 MPSSLF2 MPSSLF3 
MPSSLF4 MPSSLF5 MPSSLF6 MPSSLF7 MPSSLF8 MPSSLF9 MPSSLF10 MPSSLF11 
MPSSLF12 MPSSLF13 MPSSLF14 MPSSLF15 MPSOTH1 MPSOTH2 MPSOTH3 MPSOTH4 
MPSOTH5 MPSOTH6 MPSOTH7 MPSOTH8 MPSOTH9 MPSOTH10 MPSOTH11 MPSOTH12 
MPSOTH13 MPSOTH14 MPSOTH15 MPSSOC1 MPSSOC2 MPSSOC3 MPSSOC4 MPSSOC5 
MPSSOC6 MPSSOC7 MPSSOC8 MPSSOC9 MPSSOC10 MPSSOC11 MPSSOC12 MPSSOC13 
MPSSOC14 MPSSOC15 FCM1 FCM2 FCM3 FCM4 FCM5 FCM6 FCM7 FCM8 FCM9 FPS1 
FPS2 FPS3 FPS4 FPS5 FPS6 FPS7 FPE1 FPE2 FPE3 FPE4 FPE5 FPC1 FPC2 
FPC3 FPC4 FDA1 FDA2 FDA3 FDA4 FORG1 FORG2 FORG3 FORG4 FORG5 FORG6 
POSAFF1 POSAFF2 POSAFF3 POSAFF4 POSAFF5 POSAFF6 POSAFF7 POSAFF8 
POSAFF9 POSAFF10 NEGAFF1 NEGAFF2 NEGAFF3 NEGAFF4 NEGAFF5 NEGAFF6 
NEGAFF7 NEGAFF8 NEGAFF9 NEGAFF10 GENHSS1 GENHSS2 GENHSS3 GENHSS4 
GENHSS5 GENHSS6 GENHSS7 GENHSS8 ACAHSS1 ACAHSS2 ACAHSS3 ACAHSS4 
ACAHSS5 ACAHSS6 ACAHSS7 ACAHSS8 ACAHSS9 ACAHSS10 SOCHSS1 SOCHSS2 
SOCHSS3 SOCHSS4 SOCHSS5 SOCHSS6 SOCHSS7 SOCHSS8 SOCHSS9 SOCHSS10 
SOCHSS11 SOCHSS12 CDEN1 CDEN2 CDEN3 CDEN4 CBDIS1 CBDIS2 CBDIS3 
CBDIS4 CMDIS1 CMDIS2 CMDIS3 CMDIS4 CACTV1 CACTV2 CACTV3 CACTV4 
CPLAN1 CPLAN2 CPLAN3 CPLAN4 CSUPP1 CSUPP2 CSUPP3 CSUPP4 CPOSRE1 
CPOSRE2 CPOSRE3 CPOSRE4 CSFBLM1 CSFBLM2 CSFBLM3 CSFBLM4 SSRAP 
SSRAW SSRAR SSATP SSATW SSATR SSGP SSGW SSGR SSSIP SSSIW SSSIR 
SSRWP SSRWW SSRWR DEPEND SC EFF MCDEP MCSC RELATED DEPNDNCE NEEDY 
CONNECT DEQ1 DEQ2 DEQ3; 
 USEOBSERVATIONS ARE (DAY NE 8); ! use data from only the first 7 days 
 USEVARIABLES ARE 
ID EVENT1 EVENT3 EVENT6 SSRAP SSATP SSGP SC MPSSOC FCM FDA 
MPSSLF FPS GENHSS ACAHSS SOCHSS CDEN CBDIS CMDIS 
CACTV CPLAN CSUPP CPOSRE NAPARC1 NAPARC2 NAPARC3 
NAPARC4 NAPARC5 PAPARC1 PAPARC2 PAPARC3; 
 CLUSTER IS ID; ! Level-2 grouping identifier 
 MISSING ARE ALL (-999); 
 BETWEEN ARE MPSSOC FCM FDA SC MPSSLF FPS; 
DEFINE: 
 
!Self-Critical Perfectionism indicators (MPSSOC, FCM, FDA, SC) 
MPSSOC = MEAN(MPSSOC1-MPSSOC15); 
FCM = MEAN(FCM1-FCM9); 
FDA = MEAN(FDA1-FDA4); 



 
!Personal Standards Perfectionism indicators (MPSSLF, FPS) 
MPSSLF = MEAN(MPSSLF1-MPSSLF15); 
FPS = MEAN(FPS1-FPS7); 
 
!Hassles indicators (GENHSS, ACAHSS, SOCHSS) 
GENHSS = MEAN(GENHSS1-GENHSS8); 
ACAHSS = MEAN(ACAHSS1-ACAHSS10); 
SOCHSS = MEAN(SOCHSS1-SOCHSS12); 
 
!Avoidant Coping indicators (CDEN, CBDIS, CMDIS) 
CDEN = MEAN(CDEN1-CDEN4); 
CBDIS = MEAN(CBDIS1-CBDIS4); 
CMDIS = MEAN(CMDIS1-CMDIS4); 
 
!Problem-Focused Coping indicators (CACTV, CPLAN, CSUPP) 
CACTV = MEAN(CACTV1-CACTV4); 
CPLAN = MEAN(CPLAN1-CPLAN4); 
CSUPP = MEAN(CSUPP1-CSUPP4); 
 
!Positive Reinterpretation (CPOSRE) 
CPOSRE = MEAN(CPOSRE1-CPOSRE4); 
 
!Negative Affect indicators (NAPARC1-NAPARC5) 
NAPARC1 = MEAN(NEGAFF1 NEGAFF2); 
NAPARC2 = MEAN(NEGAFF5 NEGAFF6); 
NAPARC3 = MEAN(NEGAFF4 NEGAFF10); 
NAPARC4 = MEAN(NEGAFF3 NEGAFF7); 
NAPARC5 = MEAN(NEGAFF8 NEGAFF9); 
 
!Positive Affect indicators (PAPARC1-PAPARC3) 
PAPARC1 = MEAN(POSAFF1 POSAFF4 POSAFF7 POSAFF10); 
PAPARC2 = MEAN(POSAFF2 POSAFF5 POSAFF8); 
PAPARC3 = MEAN(POSAFF3 POSAFF6 POSAFF9); 
 
ANALYSIS: TYPE = TWOLEVEL RANDOM; 
MODEL: ! model specification follows 
 
%WITHIN% ! Model for Within effects follows 
 
!measurement model 
HASSLESw BY GENHSS ACAHSS SOCHSS; 
AVOCOPEw BY CDEN CBDIS CMDIS; 
STRESSw BY EVENT1 EVENT3 EVENT6; 
PERCSSw BY SSRAP SSATP SSGP; 
PROCOPEw BY CACTV CPLAN CSUPP; 
CPOSREw BY CPOSRE@1; 
NEGAFFw BY NAPARC1-NAPARC5; 



POSAFFw BY PAPARC1-PAPARC3; 
 
!within residual variances 
GENHSS ACAHSS SOCHSS; 
CDEN CBDIS CMDIS EVENT1 EVENT3 EVENT6 SSRAP SSATP; 
CACTV CPLAN CSUPP CPOSRE@0 NAPARC1-NAPARC5 PAPARC1-PAPARC3; 
 
!variances and covariances 
HASSLESw AVOCOPEw STRESSw PERCSSw PROCOPEw CPOSREw; 
AVOCOPEw WITH PERCSSw CPOSREw; 
PERCSSw WITH CPOSREw; 
POSAFFw NEGAFFw; POSAFFw WITH NEGAFFw; 
 
!path coefficients 
HASSLESw ON AVOCOPEw(pw); 
STRESSw ON AVOCOPEw(qw); 
PROCOPEw ON PERCSSw(rw) 
 CPOSREw(sw); 
NEGAFFw ON HASSLESw(fw) 
 AVOCOPEw(hw) 
 STRESSw(jw) 
 PERCSSw(lw); 
POSAFFw ON HASSLESw(gw) 
 AVOCOPEw(iw) 
 STRESSw(kw) 
 PERCSSw(mw) 
 PROCOPEw(nw) 
 CPOSREw(ow); 
 
%BETWEEN% ! Model for Between effects follows 
 
!measurement model 
SCPERF BY MPSSOC FCM FDA SC; 
PSPERF BY MPSSLF FPS; 
HASSLESb BY GENHSS ACAHSS SOCHSS; 
AVOCOPEb BY CDEN CBDIS CMDIS; 
STRESSb BY EVENT1 EVENT3 EVENT6; 
PERCSSb BY SSRAP SSATP SSGP; 
PROCOPEb BY CACTV CPLAN CSUPP; 
CPOSREb BY CPOSRE@1; 
NEGAFFb BY NAPARC1-NAPARC5; 
POSAFFb BY PAPARC1-PAPARC3; 
 
!between residual variances 
MPSSOC FCM FDA SC MPSSLF FPS GENHSS ACAHSS SOCHSS; 
CDEN CBDIS CMDIS EVENT1 EVENT3 EVENT6 SSRAP SSATP SSGP 
CACTV CPLAN CSUPP CPOSRE@0 NAPARC1-NAPARC5 PAPARC1-PAPARC3; 
 



!variances and covariances 
SCPERF PSPERF CPOSREb; 
SCPERF WITH PSPERF CPOSREb; 
PSPERF WITH CPOSREb; 
HASSLESb AVOCOPEb STRESSb PERCSSb PROCOPEb CPOSREb; 
POSAFFb NEGAFFb; POSAFFb WITH NEGAFFb; 
 
!path coefficients 
AVOCOPEb ON SCPERF(bb); 
HASSLESb ON SCPERF(ab) 
 AVOCOPEb(pb); 
STRESSb ON SCPERF(cb) 
 AVOCOPEb(qb); 
PERCSSb ON SCPERF(db); 
PROCOPEb ON PSPERF(eb) 
 PERCSSb(rb) 
 CPOSREb(sb); 
NEGAFFb ON HASSLESb(fb) 
 AVOCOPEb(hb) 
 STRESSb(jb) 
 PERCSSb(lb); 
POSAFFb ON HASSLESb(gb) 
 AVOCOPEb(ib) 
 STRESSb(kb) 
 PERCSSb(mb) 
 PROCOPEb(nb) 
 CPOSREb(ob); 
 
MODEL CONSTRAINT: ! compute Between and Within indirect effects 
new(baf bag bbh bbi bbp bpf bbpf bpg bbpg bbq bqj bbqj bqk bbqk 
bcj bck bdl bdm bdr brn bdrn ben bsn wpf wpg wqj wqk wrn wsn); 
 
baf = ab*fb; 
bag = ab*gb; 
bbh = bb*hb; 
bbi = bb*ib; 
bbp = bb*pb; 
bpf = pb*fb; 
bbpf = bb*pb*fb; 
bpg = pb*gb; 
bbpg = bb*pb*gb; 
bbq = bb*qb; 
bqj = qb*jb; 
bbqj = bb*qb*jb; 
bqk = qb*kb; 
bbqk = bb*qb*kb; 
bcj = cb*jb; 
bck = cb*kb; 



bdl = db*lb; 
bdm = db*mb; 
bdr = db*rb; 
brn = rb*nb; 
bdrn = db*rb*nb; 
ben = eb*nb; 
bsn = sb*nb; 
wpf = pw*fw; 
wpg = pw*gw; 
wqj = qw*jw; 
wqk = qw*kw; 
wrn = rw*nw; 
wsn = sw*nw; 
 


